

Creating SAS® Report Outline Using the Output Delivery System for Online Review

Sue X. Zhang, Abbott Nutrition Research and Development, Abbott Laboratories, Columbus, Ohio

ABSTRACT

The destination function of the output delivery system (ODS) RTF or PDF in SAS® permits the user to create linked table of contents (TOC) pages or bookmarks that, with a single mouse click, can take the user directly to the desired SAS output. It allows the user to easily move around the document online via the TOC pages or the bookmarks instead of going back and forth and scrolling through the long SAS report.

This paper presents step-by-step procedures for applying ODS CONTENTS and TOC_DATA options and ODS PROCLABEL statement to generate the informative TOC pages and the bookmarks about the SAS report for fast online review. Methods are detailed for

1. Generating the table titles in the TOC pages and the Document Map in Word
2. Presenting the data set names in the TOC pages and the Document Map in Word for the data set documentation
3. Displaying the discrepancy notices in the bookmarks in a PDF file for the two data sets compared using PROC COMPARE during the clinical study validation process.

Data presented in this paper are either directly obtained or modified from SAShelp.class for demonstration purposes only. PC SAS version 9.2 and Microsoft® Office Word 2003 on Windows XP Professional version 2002 were used in preparing this paper.

INTRODUCTION

Online document review provides advantages over traditional paper review. The document can be retrieved easily. The content is easy to navigate and search. It is environmentally friendly because it avoids paper consumption.

A SAS output report could include hundreds of tables. Going online and looking through the long document to locate the right table can be time consuming. Inserting the report outline in either the Table of Contents, in the Document Map of a Word file or in the bookmarks of a PDF file could help the reviewers to navigate the SAS report online quickly.

METHODS

DEMONSTRATION 1: GENERATING THE TABLE TITLES IN THE TOC PAGES AND THE DOCUMENT MAP IN WORD

With ODS RTF CONTENTS option and ODS PROCLABEL and CONTENTS='.....' statements (Ensor et al. 2006), a TOC can provide more informative headings about the tables generated. With a single mouse click, the linked TOC pages can take the user, directly to the desired output. Zhang (2010) has detailed the process on how to generate a TOC with informative table titles in RTF file in her paper.

In addition to TOC, another tool to facilitate online review is to use Word Document Map which allows the user to navigate the report outline quickly. To get Document Map to display the table titles in different levels, use the WORDSTYLE option following ODS RTF (<http://support.sas.com/md/base/ods/odsrtf/rtf901.html>).

The WORDSTYLE= option associates the RTF markup code \s1 with Word's Heading 1 style (Axelrod 2004).

The following code uses the WORDSTYLE option to assign the name s1 to Heading 1, s2 to Heading 2 and s3 to Heading 3. Then the names s1, s2 or s3 are used to display the table titles in different levels in Word Document Map using STYLE={PRETEXT=' \R\s1.....'}, STYLE={PRETEXT=' \R\s2.....'} or STYLE={PRETEXT=' \R\s3.....'}

For detail, see the following code:

```
options nodate nonumber byline orientation=portrait;
ods escapechar="^";
ods listing close;
ods rtf file="C:\SAS presentations\Pharma2012\Output\WordDocMapDemo.rtf"
    startpage=yes

/* The CONTENTS and TOC_DATA options generate TOC for SAS 9.2 */
contents device=SASEMF toc_data

/* The WORDSTYLE option assigns the name s1 to Heading 1, s2 to Heading 2 and s3 to
Heading 3 */
wordstyle="{\s1 Heading 1 \s2 Heading 2 \s3 Heading 3}";

/* ODS PROCLABEL '.....' generates the 1st level of the title from TOC */
ods proclabel 'Demo 1: PROC PRINT - Customize Toc and Style Template';

/* CONTENTS ='...' generates the 2nd level of the title from TOC. It can be omitted by
specifying CONTENTS=' ' */
proc print data=sashelp.class noobs label contents ='Print 2nd label'
style(report)={font_weight=bold
    pretext='\R\s1Demo 1: PROC PRINT - Customize Toc and Style Template\line\
        Print 2nd label'
};
var name /style={font_weight=bold cellwidth=3cm just=1};
var sex/style={cellwidth=3cm just=1};
run;

/* Manipulate SASHELP.CLASS data set for PROC REPORT */
data class;
set sashelp.class;
cnt=1;
run;
ods proclabel 'Demo 1: PROC REPORT - Customize Toc and Style Template';
proc report data=class nowd headskip missing contents ='Report 2nd label'

/* PRETEXT='...' places the table titles in the body of the table instead of in Word
Header (Zhang 2010) */
style(report)={font_weight=bold
    pretext='\R\s2Demo 1: PROC REPORT - Customize Toc and Style Template
        \line\ Report 2nd label\line\ Report 3rd label'
};
column cnt name sex;
define cnt / group noprint;
define name /order order=internal style(column)={font_weight=bold
    cellwidth=3cm just=left};
define sex /style(column)={cellwidth=3cm just=1};
break before cnt / contents="Report 3rd label" page;
run;

ods proclabel 'Demo 1: PROC TABULATE - Customize Toc and Style Template';
proc tabulate data=sashelp.class missing contents ='Tabulate 2nd label' ;
class sex;

/* CONTENTS ='...' generates the 3rd level of the title from TOC. It can be omitted by
specifying CONTENTS=" " */
table sex all='Total' / contents ='Tabulate 3rd label'
style={font_weight=bold
```

```


pretext='\R\s3Demo 1: PROC TABULATE - Customize Toc and Style Template
\line\ Tabulate 2nd label\line\ Tabulate 3rd label'
};
run;

ods rtf close;
ods listing;

```

To view the table titles in Document Map, go to View>>Document Map in the Word file. The output shows the table titles in the TOC page and in the Document Map. When clicking on a heading in the Document Map, the cursor will jump to that heading with the associated table. It helps to navigate the long document quickly. The heading in the Document Map can expand or collapse by clicking on the minus signs next to the titles.

Output 1: Titles in the TOC Pages and the Document Map in Word



For how to use the Document Map in Microsoft Word, see <http://www.shaunakelly.com/word/numbering/documentmap.html>.

DEMONSTRATION 2: PRESENTING THE DATA SET NAMES IN THE TOC PAGES AND THE DOCUMENT MAP IN WORD FOR THE DATA SET DOCUMENTATION

```

/* Use dictionary tables to get data set information from a library SASHELP */
proc sql;
create table _dicttbl as select *
  from dictionary.tables
  where libname="SASHELP" and memtype='DATA';

create table _diccol as
select MEMNAME 'Dataset', NAME 'Variable',
       TYPE, npos, varnum, length, format, informat, label
from dictionary.columns
where UPCASE(libname)="SASHELP";

create table _dicttblmember as select *
  from dictionary.members
  where libname="SASHELP" and memtype='DATA';
quit;
data datadoc;
merge _dicttbl _diccol _dicttblmember;
by MEMNAME;

```

```

    label memname='Dataset';
run;

options date number byline orientation=portrait;
ods escapechar="^";
ods listing close;

/* The BODYTITLE option is used here to display data set names in the Word Document
Map */
ods rtf file='C:\SAS presentations\Pharma2012\Output\Datadoc.rtf'
    contents device=SASEMF toc_data bodytitle contents;
ods proclabel 'Data Sets Documentation';
proc sort data=datadoc;


/* The BY statement is used to display data set names in the TOC pages and in the Word
Document Map */
by memname memlabel;
run;

ods proclabel "Dataset Documentation";
title;
proc report data=datadoc nowd headskip missing split='~' contents=''
    style(report)=[font=("Arial", 9pt)]
    style(Summary)=[FONT=("Arial", 9PT)]
    style(header)=[font_weight=bold FONT=("Arial", 9pt)]
    style(column)=[FONT=("Arial", 9PT)];
by memname ;
column libname name type length format label;
define libname / group noprint;
define name /style(column)={cellwidth=2.5cm just=1};
define type/style(column)={cellwidth=1.8cm just=1};
define length/display style(column)={cellwidth=1.8cm just=left};
define format/display style(column)={cellwidth=2cm just=left};
define label/style(column)={cellwidth=6cm just=left};
break before libname / contents="" page;
run;

ods _all_ close;
ods listing;


```

Output 2: Data Set Names in the Word Document Map and the TOC of the Word File



Output 3: Data Set Names in the Word Document Map and the TOC of the Word File – The Table Output

For example, by clicking dataset = CARS from either Document Map or TOC, you will see the table for data set CARS.



DEMONSTRATION 3: DISPLAYING THE DISCREPANCY NOTICES IN THE BOOKMARKS IN A PDF FILE FOR THE TWO DATA SETS COMPARED USING PROC COMPARE DURING THE CLINICAL STUDY VALIDATION PROCESS.

Double programming is commonly used for clinical study validation. The program validation consists of the independently written programs which perform the same tasks as the primary programs. PROC COMPARE is used to identify inequalities and discrepancies between the two statistical analysis data sets: the primary data sets and the validation data sets.

The following code shows how to use PROC COMPARE to verify if the two sets of data are identical

```
/* Create the library for primary and validation data sets */
libname primary 'C:\SAS presentations\Pharma2012\Data\Primary';
libname valid 'C:\SAS presentations\Pharma2012\Data\Validation';

/* Get the two identical data sets from SASHELP.CLASS */
data primary.class
    valid.class;
    set sashelp.class;
run;

options nodate nonumber byline orientation=landscape;
ods escapechar="^";
ODS LISTING CLOSE;

/* Open the PDF destination and specify the output file name and location */
/* CONTENTS option generates TOC */
ods pdf file="c:\sas presentations\pharma2012\output\compare.pdf"
    startpage=yes contents;

proc sort data=primary.class out=primary_class;
by name;
run;

proc sort data=valid.class out=val_class;
by name;
run;



ods proclabel "Listing 1: Dataset: class";
title1 j=1 "^S={leftmargin=7in}page ^{thispage} of ^{lastpage}";

title3 "Listing 1: Dataset class";

proc compare base=primary_class compare=val_class
    listall method=absolute criterion=1e-12;
    id name;
run;
ods _all_ close;
ods listing;
```

Output 4: The Output from PROC COMPARE for the Two Identical Data Sets

The two data sets are identical. The second level of title in the bookmark and TOC shows *Datasets* and *Summary*.




What if the two data sets have unequal value?

```
data primary.class ;
set sashelp.class;
```

```
run;
data valid.class;
  set sashelp.class;
  if name='John' then age=14;
run;
```

Output 5: The Output from PROC COMPARE for the Two Data Sets with Unequal Values

The two data sets have unequal values. The second level of title in the bookmark and TOC shows *Differences*.




What about the two data sets have different observations?

```
data primary.class ;
  set sashelp.class;
run;
data valid.class;
  set sashelp.class;
  if name='John' then delete;
run;
```

Output 6: The Output from PROC COMPARE for the Two Data Sets with Different Observations

The two data sets have different observations. The second level of title in the bookmark and TOC shows *Details*.




What if the two data sets have different variables?

```
/* With additional variables */
data primary.class ;
  set sashelp.class;
run;
data valid.class;
  set sashelp.class;
  hdc=45;
run;
```

Output 7: The Output from PROC COMPARE for the Two Data Sets with Different Variables

The two data sets have different variables. The second level of title in the bookmark and TOC shows *Variables*.



What if the two data sets have different attributes, i.e. different format or labels?

```


/* With different attributes - formats and labels */
proc format;
value $gender 'F' = 'female'
              'M' = 'male';
run;

DATA primary.CLASS ;
  SET sashelp.CLASS;
RUN;
DATA VALID.CLASS;
  SET sashelp.CLASS;
  format sex gender.;
  label age='Age in days';
RUN;

```

Output 8: The Output from PROC COMPARE for the Two Data Sets with Different Attributes

The two data sets have different attributes. The second level of title in the bookmark and TOC shows *Variables*.



CONCLUSIONS

Using the ODS RTF CONTENTS and TOC_DATA options and the ODS PROCLABEL statement, programmers can generate the SAS report output with a table of contents and bookmarks. TOC pages or bookmarks provide information concerning the kinds of tables produced, creating significant convenience for the reader. Specific tables can be viewed by clicking the title of the table as it appears in the TOC or the bookmarks, thus alleviating the need for lengthy scrolling through many pages of tables.

REFERENCES

- Axelrod, Elizabeth. 2004. "Skinning the Cat This Way and That: Using ODS to Create Word Documents That Work for You." SUGI 29 Proceedings. Montréal, Canada.
- Eberhardt, Peter. 2006. "How Do I Look it Up If I Cannot Spell It: An Introduction to SAS® Dictionary Tables." SUGI 31 Conference Proceedings: San Francisco, California.
- Ensor, Michele and Zender, Cynthia. 2006. Advanced Output Delivery System (ODS) Topics, Course Notes for Live Web training on 5/22/06 – 5/25/06. Cary, NC: SAS® Institute Inc.
- Lund, Pete. "A Quick and Easy Data Dictionary Macro." SAS Conference Proceedings: SUGI 27. April 2002.
- Shauna, Kelly. "How to use the Document Map in Microsoft Word." Available at <http://www.shaunakelly.com/word/numbering/documentmap.html> Accessed July 2011.
- Zhang, Sue X. 2010. "Creating Customized Word Tables with Table of Contents using SAS ODS RTF for Clinical Studies." PharmaSUG Proceedings. Orlando, Florida.

ACKNOWLEDGEMENTS

I would like to thank my manager Wendy Jones for her support. Thank Carol Richardson from our company for her time reviewing the paper. I would like to acknowledge the professional support and advice from the SAS support team. I appreciate the help from Michele Ensor from SAS during and after the Live Web training on Advanced Output Delivery System (ODS) Topics. Thanks go to everyone in my company for their help with the paper submission process.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Please contact the authors at:

Sue X. Zhang
Abbott Nutrition Research and Development
Dept. 105400, Bldg. RP3-2
3300 Stelzer Road
Columbus, OH 43219
Work Phone: 614-624-6531
Email: Sue.zhang@abbott.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Microsoft® Office is a registered trademark of the Microsoft Corporation.

Other brand and product names are registered trademarks or trademarks of their respective companies.